Solving beam deflection problems using the moment-deflection approach and using the Euler-Bernoulli approach

by Nasser M. Abbasi
November 2009

- **Links**

PDF file
Mathematica notebook

Introduction

These are problems in beam deflection showing how to use *Mathematica* to solve them.

Problem 1

This is problem 9-3, page 551, from bok Problem Solvers, strength of materials and mechanics of materials by REA. I show here how to solve this problem using *Mathematica*.

Start by setting up the moment deflection equation for the Euler beam $E\,I\;\; y''[x] = M(x)$, this equation is found for both halves of the beam resulting in 2 solutions, $y_1(x)$ and $y_2(x)$.

On the right side, boundary condition is that $y_2(L) = 0$ and on the left side $y_1(0) = 0$, then we need an additional 2 boundary conditions, for this, use the continuity conditions at $L/2$ where we set $y_1(L/2) = y_2(L/2)$ and $(y')_1\,(L/2) = (y')_2\,(L/2)$ and now we have the complete solutions.

For the left half of beam, it is easily found that $M(x) = \frac{w\,L}{8}\,x$ and for the right half $M(x) = \frac{w\,l\,x}{8} - \frac{w}{2}\left(x - \frac{l}{2}\right)^2$



- **Find $y_1(x)$**

```
Clear[E , I, w, L, eq1, eq2, y1, y2, M1, M2]
```

```
M1 = (w L/8) x;
eq1 = E I y''[x] == M1;
sol1 = First@DSolve[eq1, y[x], x];
y1 = y[x] /. sol1;
y1 = y1 /. {C[1] → c1, C[2] → c2}
```

$$c1 + c2\, x + \frac{L\, w\, x^3}{48\, E\, I}$$

- **Use first boundary condition $y_1(0) = 0$**

```
First@Solve[(y1 /. x → 0) == 0, {c1, c2}];
y1 = y1 /. %
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$c2\, x + \frac{L\, w\, x^3}{48\, E\, I}$$

- **Find $y_2(x)$**

```
M2 = (w L/8) x - (w/2) (x - L/2)^2;
eq2 = E I y''[x] == M2;
sol2 = First@DSolve[eq2, y[x], x];
y2 = y[x] /. sol2;
y2 = y2 /. {C[1] → c3, C[2] → c4}
```

$$c3 + c4\, x - \frac{L^2\, w\, x^2}{16\, E\, I} + \frac{5\, L\, w\, x^3}{48\, E\, I} - \frac{w\, x^4}{24\, E\, I}$$

- **Use boundary condition $y_2(L) = 0$**

```
First@Solve[(y2 /. x → L) == 0, {c3, c4}];
y2 = y2 /. %;
y2 = Collect[y2, c4]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

$$c4\, (-L + x) - \frac{L^2\, w\, x^2}{16\, E\, I} + \frac{5\, L\, w\, x^3}{48\, E\, I} - \frac{w\, x^4}{24\, E\, I}$$

We see from the above, that we have 2 additional constants of integrations to solve for. c2 and c4. To solve for these, we use the continuity conditions at $L/2$

```
sol = First@Solve[{(y2 == y1 /. x → L/2), (D[y2, x] == D[y1, x]) /. x → L/2}, {c2, c4}]
```

$$\left\{c2 \to -\frac{7\, L^3\, w}{384\, E\, I},\ c4 \to \frac{L^3\, w}{384\, E\, I}\right\}$$

All the integration of constants now found, we can now plot moment diagrams, shear diagram, and deflection diagrams

```
y1 = y1 /. sol
```

$$-\frac{7 L^3 w x}{384 E I}+\frac{L w x^3}{48 E I}$$

```
y2 = y2 /. sol
```

$$-\frac{L^2 w x^2}{16 E I}+\frac{5 L w x^3}{48 E I}-\frac{w x^4}{24 E I}+\frac{L^3 w (-L+x)}{384 E I}$$

- **Assign some values for E, I, L and w and do the plots. PLot the deflection**

```
L = 10;
values = {E → 2 * 10^6, I → 2, w → 10000};
Plot[{y1 HeavisideTheta[L / 2 - x] /. values, y2 HeavisideTheta[x - L / 2] /. values},
 {x, 0, 10}, Frame → True, FrameLabel → {{"y(x)", None}, {"x", "beam deflection"}}]
```



- **Plot the bending moment**

```
Plot[{M1 HeavisideTheta[L / 2 - x] /. values, M2 HeavisideTheta[x - L / 2] /. values},
 {x, 0, 10}, Frame → True, FrameLabel → {{"M(x)", None}, {"x", "Bending moment"}}]
```



- **Plot the shear diagram**

```
Plot[Evaluate[{D[M1, x] HeavisideTheta[L / 2 - x], D[M2, x] HeavisideTheta[x - L / 2] } /. values],
 {x, 0, 10}, Frame → True, FrameLabel → {{"V(x)", None}, {"x", "shear diagram"}}]
```



- **Solving the same problem directly from the Euler-Bernoulli 4th order ODE**

The above approach (using the Moment-deflection ODE) is a standard approach to solve deflection beam problems. However, we can also use the 4th order Euler beam equation direclty as follows. One needs to make sure that the load on the RHS of this ODE is the load per unit length only, i.e. w in this problem. Notice the use of the unit step function since w only acts on half the beam.

```
Clear[E, I, w, L, eq, y]
eq = E I y''''[x] == -w HeavisideTheta[x - L/2];
sol = First@DSolve[{eq, y[0] == 0, y[L] == 0, y''[0] == 0, y''[L] == 0}, y[x], x];
y = y[x] /. sol
```

$$\frac{1}{2304\, E\, I}\Big(32\, L^4\, w\, x\, \text{DiracDelta}[L] - 96\, L^3\, w\, x^2\, \text{DiracDelta}[L] + 64\, L^2\, w\, x^3\, \text{DiracDelta}[L] + 6\, L^4\, w\, \text{HeavisideTheta}\left[-\frac{L}{2}\right] - 102\, L^3\, w\, x\, \text{HeavisideTheta}\left[-\frac{L}{2}\right] + 144\, L^2\, w\, x^2\, \text{HeavisideTheta}\left[-\frac{L}{2}\right] - 48\, L\, w\, x^3\, \text{HeavisideTheta}\left[-\frac{L}{2}\right] - 42\, L^3\, w\, x\, \text{HeavisideTheta}\left[\frac{L}{2}\right] + 48\, L\, w\, x^3\, \text{HeavisideTheta}\left[\frac{L}{2}\right] - 6\, L^4\, w\, \text{HeavisideTheta}\left[-\frac{L}{2}+x\right] + 48\, L^3\, w\, x\, \text{HeavisideTheta}\left[-\frac{L}{2}+x\right] - 144\, L^2\, w\, x^2\, \text{HeavisideTheta}\left[-\frac{L}{2}+x\right] + 192\, L\, w\, x^3\, \text{HeavisideTheta}\left[-\frac{L}{2}+x\right] - 96\, w\, x^4\, \text{HeavisideTheta}\left[-\frac{L}{2}+x\right] - 2\, L^5\, w\, x\, \text{DiracDelta}'\left[-\frac{L}{2}\right] + 3\, L^4\, w\, x^2\, \text{DiracDelta}'\left[-\frac{L}{2}\right] - L^3\, w\, x^3\, \text{DiracDelta}'\left[-\frac{L}{2}\right] - L^5\, w\, x\, \text{DiracDelta}'\left[\frac{L}{2}\right] + L^3\, w\, x^3\, \text{DiracDelta}'\left[\frac{L}{2}\right]\Big)$$

- **Assign some values for E, I, L and w and do the plots. PLot the deflection**

```
L = 10;
values = {E → 2 * 10^6, I → 2, w → 10000};
Plot[y /. values, {x, 0, 10}, Frame → True,
 FrameLabel → {{"y(x)", None}, {"x", "beam deflection"}}]
```



We see that we obtain the same result as earlier.

- **Plot the bending moment**

```
Plot[Evaluate[E I D[D[y, x], x] /. values], {x, 0, 10},
 Frame → True, FrameLabel → {{"M(x)", None}, {"x", "bending moment"}}]
```



- **Plot the shear diagram**

```
Plot[Evaluate[E I D[D[D[y, x], x], x] /. values], {x, 0, 10},
 Frame → True, FrameLabel → {{"V(x)", None}, {"x", "shear diagram"}}]
```



Problem 2

In this problem. I set up a cantilever beam and first solved it using the Java beam applet that can be accessed here and then I solved this problem using Mathematica below, showing that I get the same answer as shown




```
Clear[E, I, w, L, eq, y, sol]
eq = E I y''''[x] == -w HeavisideTheta[L/2 - x];
```

The boundary conditions to use are: $y[L] == 0$, $y''[0] == 0$, $y'''[0] == 0$, $y'[L] == 0$

```
sol = Simplify@First@DSolve[{eq, y[L] == 0, y''[0] == 0, y'''[0] == 0, y'[L] == 0}, y[x], x];
y = y[x] /. sol
```

Out[8]=

$$\frac{1}{384\,E\,I} w \left(-48 L^4 + 64 L^3 x - 16 x^4 + 8 L (L - x)^2 (5 L + 4 x) \,\text{HeavisideTheta}\left[-\frac{L}{2}\right] + L^3 (7 L - 8 x) \,\text{HeavisideTheta}\left[\frac{L}{2}\right] + L^4 \,\text{HeavisideTheta}\left[-\frac{L}{2} + x\right] - 8 L^3 x \,\text{HeavisideTheta}\left[-\frac{L}{2} + x\right] + 24 L^2 x^2 \,\text{HeavisideTheta}\left[-\frac{L}{2} + x\right] - 32 L x^3 \,\text{HeavisideTheta}\left[-\frac{L}{2} + x\right] + 16 x^4 \,\text{HeavisideTheta}\left[-\frac{L}{2} + x\right]\right)$$

- **Assign same values as shown above for E, I, L and w and do the plots.**

In[17]:=
```
L = 10;
values = {E → 28 * 10^6, I → 2, w → 1};
Plot[y /. values, {x, 0, 10}, Frame → True,
 FrameLabel → {{"y(x)", None}, {"x", "beam deflection"}}]
Print["Max deflection is ", (y /. values) /. x → 0 // N, " inches"]
```

Out[19]=




```
Max deflection is -0.0000190662 inches
```

- **Plot the bending moment**

In[23]:=
```
Plot[Evaluate[E I D[D[y, x], x] /. values], {x, 0, 10},
 Frame → True, FrameLabel → {{"M(x)", None}, {"x", "bending moment"}}]
Print["Max moment is ", (E I D[D[y, x], x] /. values) /. x → 10 // N]
```

Out[23]=




```
Max moment is -37.5
```

■ Plot the shear diagram

In[25]:=
```
Plot[Evaluate[E I D[D[D[y, x], x], x] /. values], {x, 0, 10},
 Frame → True, FrameLabel → {{"V(x)", None}, {"x", "shear diagram"}}]
```

Out[25]=